package djboom.materials.display.page {
	import djboom.materials.Work;
	import djboom.materials.core.BitmapDataCore;
	import djboom.materials.display.elements.Scroll;
	import djboom.materials.display.elements.RoundedButton;
	import djboom.materials.events.PageEvent;

	import org.casalib.display.CasaBitmap;
	import org.casalib.display.CasaSprite;

	import flash.events.Event;
	import flash.events.MouseEvent;

	/**
	 * Понятия средней, истинной плотности и пористости.
	 * <br>
	 * 
	 * @author DJ BooM 
	 * @since 06.04.2010
	 * 
	 */

	public class Concepts extends Page {

		// ----------------------------------------------------------------------------
		// Константы класса.
		// ----------------------------------------------------------------------------

		// ----------------------------------------------------------------------------
		// Свойства класса.
		// ----------------------------------------------------------------------------

		private var _simpleButton:RoundedButton;
		private var _contentBitmap:CasaBitmap;
		private var _scrollContent:Scroll;
		private var _contentCase:CasaSprite;

		// ----------------------------------------------------------------------------
		// Конструктор.
		// ----------------------------------------------------------------------------

		public function Concepts() {
			super();
		}	 

		// ----------------------------------------------------------------------------
		// Методы класса.
		// ----------------------------------------------------------------------------

		private function changePage(event:MouseEvent):void {
			var pageEvent:PageEvent = new PageEvent(PageEvent.CHANGE);
			
			var enableTest:String = XML(XMLList(Navigation.NAVIGATION_XML.child("item").(@page == Page.TEST))[0]).attribute("enable").toString();
            
			if (enableTest == "false") {
				pageEvent.pageName = Page.NAVIGATION;   
			} else {
				pageEvent.pageName = Page.TEST;	
			}
			            
			dispatchEvent(pageEvent);
		}

		// ----------------------------------------------------------------------------
		// Публичные методы.
		// ----------------------------------------------------------------------------

		override public function init():void {
			// Создаем заголовок страницы.
			createTitlePage("Понятия средней, истинной плотности и пористости");
            
			_contentCase = new CasaSprite();
			_contentCase.x = 40;
			_contentCase.y = 111;
			addChild(_contentCase);
            
			_contentBitmap = new CasaBitmap(BitmapDataCore.getBitmapData(BitmapDataCore.PAGE_CONCEPTS));
			_contentCase.addChild(_contentBitmap);
			
			_contentCase.graphics.beginFill(0xff0000, 0);
			_contentCase.graphics.drawRect(0, 0, _contentBitmap.width, _contentBitmap.height);
			_contentCase.graphics.endFill();
			
			_scrollContent = new Scroll(_contentCase, 530, 530);
			_scrollContent.x = 990;
			_scrollContent.y = 107;
			addChild(_scrollContent);
             
			_simpleButton = new RoundedButton("Продолжить");
			_simpleButton.x = Work.WIDTH * 0.5 - _simpleButton.width * 0.5;
			_simpleButton.y = Work.HEIGHT - 118;
			_simpleButton.addEventListener(MouseEvent.CLICK, changePage);
			addChild(_simpleButton);
		}

		override protected function clearPage(event:Event):void {
			super.clearPage(event);
			
			_simpleButton.destroy();
			_simpleButton = null;
            
			_scrollContent.destroy();
			_scrollContent = null;
            
			_contentBitmap.destroy();
			_contentBitmap = null;
		}
        
		// ----------------------------------------------------------------------------
        // Публичные свойства.
        // ----------------------------------------------------------------------------
	}
}
